

## 10.1 One-Dimensional Real Fourier Transform

### A. Purpose

This subroutine computes discrete Fourier transforms for real data using the fast Fourier transform. The relations between the real values  $x$  and the complex Fourier coefficients  $\xi$  are as follows:

For Fourier analysis:

$$\xi_k = \frac{1}{N} \sum_{j=0}^{N-1} x_j W^{-jk}, \quad k = 0, 1, \dots, N-1, \quad (1)$$

For Fourier synthesis:

$$x_j = \sum_{k=0}^{N-1} \xi_k W^{jk}, \quad j = 0, 1, \dots, N-1, \quad (2)$$

where  $N = 2^M$  and  $W = e^{2\pi i/N} = \cos(2\pi/N) + i \sin(2\pi/N)$ , and  $i^2 = -1$ . Note that the sign on the exponent of  $W$  is negative for analysis and positive for synthesis.

### B. Usage

#### B.1 Program Prototype, Single Precision

**INTEGER** M, MS

**REAL** A( $\geq 2^M$ ), S( $\geq 2^{M-2} - 1$ )

**CHARACTER** MODE

On the initial call set MS = 0 to indicate the array S() does not yet contain a sine table. Assign values to A(), MODE, and M.

**CALL SRFT1(A, MODE, M, MS, S)**

On return A() will contain the computed results. S() will contain the sine table used in computing the Fourier transform. MS may have been changed.

#### B.2 Argument Definitions

**A()** [inout] If the argument MODE selects Analysis, A() must contain  $x$  on input and will contain  $\xi$  on output. If MODE selects Synthesis, A() must contain  $\xi$  on input and will contain  $x$  on output. When A() contains  $x$ 's,  $A(k+1)$  contains  $x_k$ ,  $k = 0, 1, \dots, N-1$ . When A() contains  $\xi$ 's, the real coefficients  $\xi_0$  and  $\xi_{N/2}$  are in A(1) and A(2), respectively and the complex coefficients,  $\xi_k$ , for  $k = 1, \dots, (N/2) - 1$ , are in the paired locations ( $A(2k+1)$ ,  $A(2k+2)$ ). The remaining complex coefficients satisfy  $\xi_{N-k} = \bar{\xi}_k$ , for  $k = 1, \dots, \frac{N}{2} - 1$ .

**MODE** [in] MODE = 'A' or 'a' selects Analysis. MODE = 'S' or 's' selects Synthesis.

**M** [in] The number of data points is  $N = 2^M$ . If M = 0, the subroutine returns taking no action. Require  $0 \leq M \leq 31$ .

**MS** [inout] Gives the state of the sine table in S(). Let  $MS_{in}$  and  $MS_{out}$  denote the values of MS on entry and return respectively. If the sine table has not previously been computed, set  $MS_{in} = 0$  or  $-1$  before the call. Otherwise the value of  $MS_{out}$  from the previous call using the same S() array can be used as  $MS_{in}$  for the current call.

Certain error conditions described in Section E cause the subroutine to set  $MS_{out} = -2$  and return. Otherwise, with  $M > 0$ , the subroutine sets  $MS_{out} = \max(M, MS_{in})$ .

If  $MS_{out} > \max(2, MS_{in})$ , the subroutine sets  $NT = 2^{MS_{out}-2}$  and fills S() with  $NT - 1$  sine values.

If  $MS_{in} = -1$ , the subroutine returns after the above actions, not transforming the data in A(). This is intended to allow the use of the sine table for data alteration before a subsequent Fourier transform, as discussed in Section G of Chapter 16.0, and as illustrated in the example of Section C.

**S()** [inout] When the sine table has been computed,  $S(j) = \sin \pi j / (2 \times NT)$ ,  $j = 1, 2, \dots, NT - 1$ , see MS above.

#### B.3 Modifications for Double Precision

Change SRFT1 to DRFT1, and the REAL type statement to DOUBLE PRECISION.

### C. Examples and Remarks

Program DRSRFT1 computes an estimate of the spectral composition of

$$f(t) = [\sin 2\pi(t + .01) + 4 \cos 2\pi(\sqrt{2}t + .3)] \quad (3)$$

with output in ODSRFT1. It is assumed  $t$  has units of seconds,  $f$  has discrete frequencies with mutual separation of at least 0.4 cycles/second, and that there is no interest in frequencies  $> 2$  cycles/second. To resolve different frequencies we select  $\Delta\omega = 0.4/10 = .04$  cycles/second, which according to Eq. (20) of Chapter 16.0 implies a sampling period of  $T = 1/\Delta\omega = 25$  seconds. According to Eq. (24) of Chapter 16.0,  $\Delta t \leq \frac{1}{2}\omega_{N/2} = \frac{1}{4}$  second, if any information about the highest frequency is to be obtained. Thus,  $N = T/\Delta t \geq 100$ . Since modest accuracy is sufficient, we give results for  $N = 128$  in the program

below. Lanczos sigma factors are used as described in Section G of Chapter 16.0. The complex transform has peaks for  $k = 26$  and for  $k = 36$ . These values of  $k$  correspond to frequencies of 1 and 1.4 cycles/second. Results for this problem, but without the use of Sigma factors are given in Chapter 16.3.

## D. Functional Description

From Eq. (1), the fact that  $x_j$  is real, and  $W^N = e^{2\pi i} = 1$ , it follows that  $\xi_{N-k} = \bar{\xi}_k$ . Thus  $\xi_{N/2}$  is real, and clearly  $\xi_0$  is also real. Thus only half the  $\xi$ 's need be stored if  $\xi_0$  and  $\xi_{N/2}$  are counted as one. The  $\xi$ 's are stored in the same space used for the  $x$ 's by taking advantage of this symmetry.

The primitive FFT routine SFFT (or DFFT) is used to compute  $\zeta$  from  $x$ , or  $x$  from  $\zeta$  where  $\zeta$  is defined by

$$\zeta_k = \frac{2}{N} \sum_{j=0}^{\frac{N}{2}-1} (x_{2j} + ix_{2j+1}) W^{-2jk}, k = 0, 1, \dots, \frac{N}{2} - 1. \quad (4)$$

Using Eq. (4), one finds that  $\zeta_k + \bar{\zeta}_{(N/2)-k}$  depends only on  $x_{2j}$  and  $\zeta_k - \bar{\zeta}_{(N/2)-k}$  only on  $x_{2j+1}$ . From this and Eq. (1) there follows immediately

$$\begin{aligned} \xi_k &= \frac{1}{4} \left[ (\zeta_k + \bar{\zeta}_{(N/2)-k}) \right. \\ &\quad \left. - i W^{-k} (\zeta_k - \bar{\zeta}_{(N/2)-k}) \right], \quad 0 < k < N/4 \\ \bar{\xi}_{(N/2)-k} &= \frac{1}{4} \left[ (\zeta_k + \bar{\zeta}_{(N/2)-k}) \right. \\ &\quad \left. + i W^{-k} (\zeta_k - \bar{\zeta}_{(N/2)-k}) \right], \quad 0 < k < N/4 \quad (5) \\ \xi_0 &= \frac{1}{2} [\Re \zeta_0 + \Im \zeta_0] \\ \xi_{N/2} &= \frac{1}{2} [\Re \zeta_0 - \Im \zeta_0] \\ \xi_{N/4} &= \frac{1}{2} \bar{\zeta}_{N/4} \end{aligned}$$

Thus to compute  $\xi$  given  $x$ ,  $\zeta/4$  is computed using Eq. (4) (divided by 4) and the  $\xi$ 's are computed from Eq. (5). To compute  $x$  given  $\xi$ , the above process is reversed. More details can be found in [1].

## References

1. Fred T. Krogh, **RFT1 — One-dimensional Real Fourier Transform**. TU Doc. CP-2310, NPO 11649, Jet Propulsion Laboratory, Pasadena, CA (1970).

## E. Error Procedures and Restrictions

M must satisfy  $0 \leq M \leq 31$  and MODE must have one of its allowed values. If these conditions are violated an error message will be issued using the error processing procedures of Chapter 19.2 with a severity level of 2 to stop execution. A return will be made with MS = -2 instead of stopping if the statement "CALL ERMSET(-1)" is executed before calling this subroutine.

If the sine table does not appear to have valid data, an error message is printed, and the sine table and then the transform are computed.

## F. Supporting Information

The source language is ANSI Fortran 77.

Entry	Required Files
<b>DRFT1</b>	DFFT, DRFT1, ERFIN, ERMSG, IERM1, IERV1
<b>SRFT1</b>	ERFIN, ERMSG, IERM1, IERV1, SFFT, SRFT1

Subroutine designed and written by: Fred T. Krogh, JPL Section 373, August 1969; Revised January 1988.

## DRSRFT1

```

c      program DRSRFT1
c>> 1996-06-05 DRSRFT1 Krogh Fixes for conversion to C.
c>> 1996-05-28 DRSRFT1 Krogh Moved formats up.
c>> 1994-10-19 DRSRFT1 Krogh Changes to use M77CON
c>> 1994-08-09 DRSRFT1 WVS Removed '0' from format
c>> 1992-04-22 DRSRFT1 CAO, commented program statement
c>> 1989-05-07 DRSRFT1 FTK, CLL
c>> 1989-05-04 DRSRFT1 FTK, CLL
c      Demo driver for SRFT1 — One-dimensional real Fourier transform
c
c-----S replaces "?: DR?RFT1, ?RFT1
c
integer I, K, KLAST, M, MS, N, N2, N4
parameter (M = 7)
c      parameter (N = 2**M)
parameter (N = 128)
parameter (N2 = N/2)
parameter (N4 = N/4)
real      F(N), S(N4-1), T, DELTAT, TTIME, SIG
real      PI, TWOPI, RIWO, ZERO, ONE, FOUR, P1, P3
real      PIN, TB
parameter (PI = 3.1415926535897932384E0)
parameter (TWOPI = 2.E0 * PI)
parameter (RIWO = 1.4142135623730950488E0)
parameter (ZERO = 0.E0)
parameter (ONE = 1.E0)
parameter (FOUR = 4.E0)
parameter (P1 = .1E0)
parameter (P3 = .3E0)
data TTIME /25.E0/
1111 format(/' Transform for K=21 to 41 using sigma factors')
2222 format(' K=', I3, ' TO', I3, 3(1P,E12.2, E10.2))
c
KLAST = N2 - 1
DELTAT = TTIME / REAL(N)
T = ZERO
c      Get sine table for use in getting sigma factors
MS = -1
call SRFT1(F, 'A', M, MS, S)
PIN = PI / REAL(N)
do 100 K = 1, KLAST
    T = T + DELTAT
    TB = TTIME - T
c
c      Get sigma factor
    if (K .GT. N4) then
        SIG = S(N2-K)
    ELSE if (K .LT. N4) then
        SIG = S(K)
    else
        SIG = ONE
    end if
    SIG = SIG / (PIN * REAL(N - 2*K))
c
c      Compute F and apply sigma factors
c
F(K+1) = SIG*(SIN(TWOPI*(T+P1)) + FOUR*COS(TWOPI*(RIWO*T+P3)))

```

```

      F(N-K+1)=SIG*(SIN(TWOPI*(TB+P1))+FOUR*COS(TWOPI*(RIWO*TB+P3)))
100 continue
      F(1) = ZERO
      T = T + DELTAT
      F(N2+1) = SIN(TWOPI*(T+P1)) + FOUR*COS(TWOPI*(RIWO*T+P3))
      call SRFT1(F, 'A', M, MS, S)
      write (*, 1111)
      do 120 K = 21, 41, 3
          write (*, 2222) K, K+2, (F(I), I = 2*K-1, 2*K+4)
120 continue
      stop
      end

```

### ODSRFT1

Transform for K=21 to 41 using sigma factors

K= 21 TO 23	-7.59E-04	1.59E-03	-1.40E-03	2.54E-03	-2.87E-03	4.67E-03
K= 24 TO 26	-7.60E-03	1.13E-02	-6.55E-02	9.12E-02	1.74E-01	-2.39E-01
K= 27 TO 29	-6.51E-02	9.11E-02	-6.77E-03	1.12E-02	-1.48E-03	4.48E-03
K= 30 TO 32	7.96E-04	2.24E-03	2.72E-03	1.10E-03	5.40E-03	2.65E-04
K= 33 TO 35	1.08E-02	-7.77E-04	2.69E-02	-3.23E-03	1.44E-01	-1.98E-02
K= 36 TO 38	-1.05E+00	1.48E-01	8.07E-01	-1.14E-01	1.91E-03	-1.74E-05
K= 39 TO 41	6.24E-03	-6.68E-04	4.21E-03	-4.10E-04	2.89E-03	-2.48E-04